version 19
drop _all

* change the root folder according to your computer directory
global root = "D:\WorkingPaper-Series\Township_Light\Replication_Files"  
     
global dofiles = "$root\dofile"     
global working_data = "$root\workdata"
global tables = "$root\table"
global figures = "$root\figure"

cd "$working_data"

******************
*****Pre Set******
******************

global geo_ctrl = "topography_t* slope_t* roughness_t*"
global wth_ctrl = "gaez_t* precipitation_t* sunlight_t*"
global dist_ctrl = "county_dist_t* dist_hz_t* dist_coast_t*"

global full_ctrl = "$geo_ctrl $wth_ctrl $dist_ctrl"

global geo_ctrl_1 = "topography_t_1 slope_t_1 roughness_t_1"
global wth_ctrl_1 = "gaez_t_1 precipitation_t_1 sunlight_t_1"
global dist_ctrl_1 = "county_dist_t_1 dist_hz_t_1 dist_coast_t_1"

global full_ctrl_1 = "$geo_ctrl_1 $wth_ctrl_1 $dist_ctrl_1"

global geo_ctrl_2 = "topography_t_2 slope_t_2 roughness_t_2"
global wth_ctrl_2 = "gaez_t_2 precipitation_t_2 sunlight_t_2"
global dist_ctrl_2 = "county_dist_t_2 dist_hz_t_2 dist_coast_t_2"

global full_ctrl_2 = "$geo_ctrl_2 $wth_ctrl_2 $dist_ctrl_2"

global matchvar = "county_dist topography slope gaez sunlight dist_hz dist_coast"


global pre_ctrl = "light_t road_t pop_t"

********************
***** Figure 5 *****
********************

use "township_panel_final.dta", clear

forvalues i = 2014 (1) 2020 {
	qui reghdfe log_light village treat_trend $full_ctrl if first_year == `i' | first_year == 0, a(county_year idcode) cl(idcode)
	local coef_`i' = e(b)[1,1]
	local se_`i' = sqrt(e(V)[1,1])
	disp `coef_`i''
	disp `se_`i''
}

mat input coef = (`coef_2014',`coef_2015',`coef_2016',`coef_2017',`coef_2018',`coef_2019',`coef_2020')
mat input std_err = (`se_2014',`se_2015',`se_2016',`se_2017',`se_2018',`se_2019',`se_2020')

mat cohort_est = [coef\std_err]
mat cohort_est = cohort_est'
svmat cohort_est

keep if cohort_est1 != .
keep cohort_est1 cohort_est2
rename (cohort_est1 cohort_est2) (coef std_err)

gen cohort = _n + 2013
gen sign = (coef > 0)

export delimited using "by_cohort_est.csv", replace
